Programming Absorptive Side Boundaries for Migration

by Robert W. Clayton

The absorptive boundary condition (B.C.) proposed by Jon Claerbout (this SEP report) is incorporated into the wave equation to eliminate side reflections which are artifacts of the grid boundary. By way of review of Claerbout's article we shall simply state that the B.C. was derived from a dispersion relation which has the functional form of a pole and a zero in k_x. This function can be made to approximate the dispersion relation of the wave equation on one side while diverging from it on the other. This property, as shown by Claerbout, gives rise to a small reflection coefficient at the boundary.

The right boundary condition in its most general form is

$$D(k_x, k_z) = (1 - c_3 v k_x/\omega)\, v_{kz}/\omega - (c_1 - c_2 v k_x/\omega) = 0 \qquad (1)$$

where c_1 , c_2 and c_3 are constants to be determined by matching the B.C. to the interior wave equation. The left B.C. is

$$D(-k_x, k_z) = 0$$

Fitting the Boundary Condition

The first step is to put $D(k_x,k_z)$ into a retarded time frame by the transform

$$\begin{cases} x' = x \\ z' = z \\ t' = t - z/v \end{cases} \Rightarrow \begin{cases} k_x = k'_x \\ k_z = k'_z + \omega/v \\ \omega = \omega' \end{cases} \qquad (2)$$

Thus,

$$D'(k_x', k_z') = (1 - c_3' \, v k_x'/\omega') \, v k_z'/\omega' \; - \; (c_1' - c_2 \, v k_x'/\omega') = 0 \tag{3}$$

where $c_1' = c_1 - 1$, $c_2' - c_2 - c_3$ and $c_3' = c_3$.

The constants are formed by matching D' to the interior wave equation (15° equation), which in the transformed frame has the form

$$v k_z/\omega = -\frac{1}{2}(v k_x/\omega)^2 \tag{4}$$

Since the reflections we seek to minimize are artifacts of the computational procedure, it is appropriate to include the effects of discretization in equations (3) and (4) . To do this we consider the effect of discretization in x only and use the following center differenced derivatives (see Lynn, SEP-8).

$$\partial_x \rightarrow \frac{2i}{\Delta x} \tan\left(m \frac{v k_x}{\omega}\right) \equiv \frac{2i}{\Delta x} T$$

$$\partial_{xx} \rightarrow \frac{1}{\Delta x^2} \frac{\delta_{xx}}{1+b\,\delta_{xx}} \rightarrow \frac{-1}{\Delta x^2} \frac{4\sin^2(m\frac{vk}{\omega})}{1-4b\sin^2(m\frac{vk}{\omega})} \equiv \frac{\frac{-4}{\Delta x^2} S^2}{1-4b\,S^2} \tag{5}$$

where $m = \dfrac{\omega \Delta x}{2v}$.

Thus, the dispersion relations of (3) and (4), in the discretized domain are respectively

$$\frac{v k_z'}{\omega} = (c_1' - c_2' \, T/m) / (1 - c_3' \, T/m)$$

$$\frac{v k_z'}{\omega} = -\frac{1}{2} S^2/m \, / \, (1 - 4b \, S^2)$$

Matching the above expressions leads to the following system for c_1' , c_2' and c_3' .

$$c_1' \left(1-4b\, S_i^2 \right) + c_2' \left(-T_i/m(1-4b\, S_i^2) \right) + c_3' \left(-\frac{1}{2} S_i^2\, T_i/m^3 \right)$$

$$= -\frac{1}{2}\, S_i^2/m^2 \ , \quad i = 1,\ 2,\ 3$$

where $\{T_i, S_i\}$ are the sine and tangent functions evaluated at particular values of (mvk_x/ω) .

The choice of fitting points is data dependent, but there are a few general guidelines to be considered. The folding point for the tangent and sine approximations is at $\pi/2m$, and consequently, the fitting points $\{x_i\}$ $(x=vk_x/\omega)$ should be less than this point. If the data are bandlimited, then $\{x_i\}$ should be less than the worst case which occurs at ω_{max} (i.e., $x \leq \frac{\pi v}{\Delta x \omega_{max}}$) . A special case exists for fitting at $x=0$. If this is done then $c_1' = 0$, which slightly simplifies the boundary algorithm by removing a time integration. However, fitting at $x=0$ is essentially a waste of one coefficient of the B.C. because the reflection coefficient at $x=0$ is always of unit magnitude, whether or not the B.C. is fitted there. This is seen from the fact that

$$\lim_{x\to 0} D' (x,\ k_z') = \lim_{x\to 0} D' (-x,\ k_z)$$

Thus the appropriate choices for $\{x_i\}$ are in the range

$$0 < x_i \leq \frac{\pi v}{\Delta x \omega_{max}}$$

Incorporating the Absorptive Boundary Conditions into the Wave Equation

Equation (3) can be rewritten as

$$[\ 1 + \frac{c_3' v(i k_x)}{(-i\omega)}\ ]\ (i k_z) \ + \ \frac{c_1'(-i\omega)}{v} + c_2'(i k_x) \ = \ 0$$

The i's have been included to enable a direct conversion to the derivative form (i.e., $ik_x \rightarrow \partial_x$) , and ω has been placed in the denominator of the first term to be compatible with the time integrated form of the wave equation (see Riley, p. 60 SEP-1).

Equation (6) is a first order differential in x , and hence, the B.C. can be represented as a linear combination of the two points at each end of the x vector. For example, on the right boundary

$$\alpha_R\ P_k^{\ j}(N-1) \ + \ \beta_R\ P_k^{\ j}(N) \ = \ RHS_R$$

where $P_k^{\ j}(n)$ is the pressure at $j\Delta t$, $k\Delta z$ and $n\Delta x$. The tridiagonal system algorithm is easily modified to incorporate (4) and the corresponding B.C. on the left side.

$$\alpha_L\ P_k^{\ j}(0) + \beta_L\ P_k^{\ j}(1) \ = \ RHS_L \qquad (8)$$

Using the notation and method given in Claerbout (Fundamentals of Geophysical Data Processing, pp. 188-189) it is necessary to specify E_0 , F_0 and T_N . The relevant recursive equations are

$$T_0 = E_0 \, T_1 + F_0 \tag{9}$$

and

$$T_{N-1} = E_{N-1} \, T_N + F_{N-1} \tag{10}$$

Comparing (8) and (9) gives

$$E_0 = \beta_L \,/\, \alpha_L \quad \text{and} \quad F_0 = RHS_L \,/\, \alpha_L \tag{11}$$

Solving (7) and (10) gives

$$T_N = (RHS_R - \alpha_R \, F_{N-1}) \,/\, (\beta_R - \alpha_R \, E_{N-1}) \tag{12}$$

To find α , β and RHS for each side, the finite differences implied by equation (6) are done. This leads to the following system

$$\begin{aligned} &\frac{1}{\Delta z} \left[\left(P^j_{k+1}(n+1) + P^j_{k+1}(n) \right) + \frac{c'_3 \, v\Delta t}{2\Delta x} \left(\frac{1+z}{1-z} \right) \left(P^j_{k+1}(n+1) - P^j_{k+1}(n) \right) \right] \\ &- \frac{1}{\Delta z} \left[\frac{1}{2} \left(P^j_k(n+1) + P^j_k(n) \right) + \frac{c'_3 \, v\Delta t}{2\Delta x} \left(\frac{1+z}{1-z} \right) \left(P^j_k(n+1) - P^j_k(n) \right) \right] \\ &+ \frac{c'_1}{2 \, v\Delta t} \left(\frac{1-z}{1+z} \right) \left[P^j_{k+1}(n+1) + P^j_{k+1}(n) + P^j_k(n+1) + P^j_k(n) \right] \\ &+ \frac{c'_2}{2\Delta x} \left[P^j_{k+1}(n+1) - P^j_{k+1}(n) + P^j_k(n+1) - P^j_k(n) \right] = 0 \end{aligned} \tag{13}$$

The bilinear transforms $\frac{\Delta t}{2}\left(\frac{1+z}{1-z}\right)$ and $\frac{2}{\Delta t}\left(\frac{1-z}{1+z}\right)$ were used to approximate $\left(\frac{1}{-i\omega} \right)$ and $(-i\omega)$ respectively.

Taking $P^{\cdot}_{\cdot}\,(n+1)$ to be the endpoint on the right and putting equation (13) in the form

$$\alpha_R \; P^j_{k+1}(n) + \beta_R \; P^j_{k+1}(n+1) = RHS_R$$

we find that

$$\alpha_R = \frac{1}{2}\left(\frac{1}{\Delta z} - \zeta - \eta + \gamma\right) \equiv \alpha$$

$$\beta_R = \frac{1}{2}\left(\frac{1}{\Delta z} + \zeta + \eta + \gamma\right) \equiv \beta$$

$$RHS_R = \mu \; P_k(n) + \nu \; P_k(n+1) - \zeta \; SUM_R + \gamma \; TERM_R$$

where

$$\zeta = \frac{c_3' \; v \; t}{\Delta z \Delta x} \qquad \eta = \frac{c_2'}{\Delta x} \qquad \gamma = \frac{c_1'}{v \Delta t}$$

$$\mu = \frac{1}{2}\left(\frac{1}{\Delta z} - \zeta + \eta - \gamma\right) \qquad \nu = \frac{1}{2}\left(\frac{1}{\Delta z} + \zeta - \eta - \gamma\right)$$

$$SUM_R = \sum_{i=1}^{\infty} \left[P^{j-i}_{k+1}(n+1) - P^{j-i}_{k+1}(n) - P^{j-i}_{k}(n+1) + P^{j-i}_{k}(n) \right]$$

$$TERM_R = \sum_{i=1}^{\infty} (-1)^{i+1} \left[P^{j-i}_{k+1}(n+1) + P^{j-i}_{k+1}(n) + P^{j-i}_{k}(n+1) + P^{j-i}_{k}(n) \right]$$

The following recursive relations can be used for the last two expressions.

$$SUM^{j+1}_R = SUM^j_R + P^j_{k+1}(n+1) - P_{k+1}(n) - P^j_k(n+1) + P^j_k(n) \quad , \qquad SUM^0_R = 0$$

$$TERM^{j+1}_R = - TERM^j_R + P^j_{k+1}(n+1) + P^j_{k+1}(n) + P^j_k(n+1) + P^j_k(n) \quad , \quad TERM^0_R = 0$$

For the left side B.C. (found by replacing x by $-x$) we have

$$\beta \; P^j_{k+1}(0) + \alpha \; P^j_{k+1}(1) = \nu \; P^j_k(0) + \mu \; P^j_k(1) + \zeta \cdot SUM_L + \gamma \cdot TERM_L$$

where SUM_L and $TERM_L$ are the same as SUM_R and $TERM_R$ with $n=0$.

Putting the above equations in an algorithmic form we have (see Riley, SEP-1, p. 60 for the wave equation part)

$k = 1 \ldots NZ$ (outer loop, enclosing all rows below)

Wave Equation	Left B.C.	Right B.C.
$\underline{u} \leftarrow \underline{0}$	SUML=0	SUMR=0
	TERML=0	TERMR=0

$j = 1 \ldots NT$ (inner loop, enclosing the rows below)

$\underline{s} \leftarrow \underline{P}_k^j$	PL=S(1),QL=S(2)	PR=S(N-1),QR=S(N)
$\underline{f} \leftarrow \underline{s} - a\,T(\underline{s}+2\underline{u})$	$RHSL = \nu\,PL + \mu\,QL$ $+ \zeta\,SUML + \gamma\,TERML$	$RHSR = \mu\,PR + \nu\,QR$ $- \zeta\,SUMR + \gamma\,TERMR$
Solve $(I + a\,T)\,\underline{P}_k^j = \underline{f}$ with left and right B.C.'s		
	$SUML \leftarrow SUML + P_k^j(2)$ $- P_k^j(1) - QL + PL$	$SUMR \leftarrow SUMR + P_k^j(N)$ $- P_k^j(N-1) - QR + PR$
$\underline{u} \leftarrow \underline{u} + P_k^j + \underline{s}$		
	$TERML \leftarrow -TERML + P_k^j(2)$ $+ P_k^j(1) + QL + PL$	$TERMR \leftarrow -TERMR + P_k^j(N)$ $+ P_k^j(N-1) + QR + PR$

As an example of the absorbing boundary conditions, a sloping wave front was diffracted outward 20 z steps by the 15° wave equation with both absorptive and reflective sides. The x-t frame is shown at every fifth z step. The example illustrates that the absorptive sides have diminished the artificial reflected energy to the point of almost being invisible.

The wave equation parameters that were used in the example are

$dx = 1$	$\omega_{max} = 2$
$dt = 1$	$x_1, x_2, x_3 = .1, .2, .4$
$dz = 8$	$b = .14868$
$v = 1$	

Accompanying the example are two tables showing the reflection coefficients at the side boundaries as a function of frequency and k_x (Table I) or angle (Table II) for the above parameters. The reflection coefficients are given as percentages and for Table II angle (θ) is defined as follows.



Following the tables, a listing of the wave equation program with absorptive side boundaries is given.



Reflecting Sides

Absorbing Sides

FREQUENCY

KX	0.5	1.0	1.5	2.0	2.5	3.0	3.5	4.0	4.5	5.0	5.5	6.0	6.5	7.0	7.5	8.0
0.00	-97	-98	-98	-98	-98	-98	-98	-98	-98	-98	-98	-98	-98	-98	-98	-98
0.10	0	0	-6	-15	-23	-29	-35	-40	-45	-48	-52	-54	-57	-59	-61	-63
0.20	-1	0	1	0	-2	-6	-10	-15	-19	-22	-26	-29	-32	-35	-38	-40
0.30	4	-1	0	1	1	0	-1	-3	-6	-9	-12	-14	-17	-20	-22	-25
0.40	13	0	-1	0	1	2	1	0	0	-2	-4	-6	-8	-10	-12	-14
0.50	20	1	-1	0	0	1	2	2	1	1	0	-1	-2	-4	-6	-7
0.60	27	5	0	-1	0	0	1	2	2	2	2	1	0	0	-1	-2
0.70	34	10	1	0	0	0	1	2	2	3	3	2	2	1	1	0
0.80	39	14	4	0	0	0	0	1	2	3	3	3	3	3	2	2
0.90	44	19	7	2	0	0	0	1	2	2	3	3	4	4	3	3
1.00	49	23	10	4	1	0	0	1	2	2	3	3	4	4	4	4
1.10	53	28	14	7	3	1	1	1	2	2	3	3	4	4	5	5
1.20	56	32	18	9	5	3	2	2	2	2	3	4	4	5	5	5
1.30	60	36	21	13	7	5	3	3	3	3	3	4	4	5	5	6
1.40	63	40	25	16	10	7	5	4	4	4	4	4	5	5	6	6
1.50	66	43	29	19	13	9	7	6	5	5	5	5	6	6	7	7
1.60	68	47	33	23	16	12	9	8	7	6	6	6	7	7	8	8
1.70	71	51	36	27	20	15	12	10	9	8	8	8	8	8	9	9
1.80	73	54	40	30	23	18	15	13	11	10	10	10	10	10	10	11
1.90	75	57	44	34	27	22	18	16	14	13	12	12	12	12	12	12
2.00	77	61	48	38	31	26	22	19	17	16	15	14	14	14	14	15
2.10	79	64	52	43	35	30	26	23	21	19	18	18	17	17	17	17
2.20	81	67	56	47	40	35	30	27	25	23	22	21	21	20	20	20
2.30	83	70	60	51	45	39	35	32	30	28	26	25	25	25	24	24
2.40	85	74	64	56	50	45	40	37	35	33	31	30	30	29	29	29
2.50	87	77	68	61	55	50	46	43	40	39	37	36	35	35	34	34
2.60	89	80	73	66	61	56	52	49	47	45	44	43	42	41	41	40
2.70	91	84	77	71	67	63	59	56	54	52	51	50	49	49	48	48
2.80	93	87	82	77	73	70	67	64	62	61	59	59	58	57	57	56
2.90	95	90	86	83	80	77	75	73	72	70	69	68	68	67	67	67
3.00	96	94	91	89	87	86	84	83	82	81	80	80	79	79	79	79
3.10	98	97	97	96	95	95	94	94	94	93	93	93	93	93	93	93

TABLE I

FREQUENCY

ANG	0.5	1.0	1.5	2.0	2.5	3.0	3.5	4.0	4.5	5.0	5.5	6.0	6.5	7.0	7.5	8.0
0.0	-98	-98	-98	-98	-98	-98	-98	-98	-98	-98	-98	-98	-98	-98	-98	-98
0.5	-72	-72	-72	-72	-72	-72	-72	-72	-72	-72	-72	-72	-72	-72	-72	-72
1.0	-53	-53	-53	-53	-53	-53	-53	-53	-53	-53	-53	-53	-53	-53	-53	-53
1.5	-39	-39	-39	-39	-39	-39	-39	-39	-38	-38	-38	-38	-38	-38	-38	-38
2.0	-28	-28	-28	-28	-28	-28	-28	-28	-28	-28	-28	-28	-27	-27	-27	-27
2.5	-19	-19	-19	-19	-19	-19	-19	-19	-19	-19	-19	-19	-19	-19	-19	-19
3.0	-13	-13	-13	-13	-13	-13	-13	-13	-13	-13	-13	-13	-13	-13	-13	-13
3.5	-9	-9	-9	-9	-9	-9	-9	-8	-8	-8	-8	-8	-8	-8	-8	-8
4.0	-5	-5	-5	-5	-5	-5	-5	-5	-5	-5	-5	-5	-5	-4	-4	-4
4.5	-3	-3	-3	-3	-3	-3	-2	-2	-2	-2	-2	-2	-2	-2	-2	-1
5.0	-1	-1	-1	-1	-1	-1	-1	-1	0	0	0	0	0	0	0	0
5.5	0	0	0	0	0	0	0	0	0	0	0	0	1	1	1	1
6.0	0	0	0	0	0	1	1	1	1	1	1	1	2	2	2	2
6.5	1	1	1	1	1	1	1	1	2	2	2	2	2	3	3	3
7.0	1	1	1	1	1	2	2	2	2	2	2	3	3	3	4	4
7.5	1	1	1	1	2	2	2	2	2	2	3	3	3	4	4	4
8.0	1	1	1	2	2	2	2	2	2	3	3	3	4	4	4	5
8.5	1	1	1	1	2	2	2	2	2	3	3	3	4	4	5	5
9.0	1	1	1	1	1	2	2	2	2	3	3	3	4	4	5	6
9.5	1	1	1	1	1	1	2	2	2	3	3	3	4	5	5	6
10.0	1	1	1	1	1	1	1	2	2	2	3	3	4	5	5	6
10.5	0	0	0	1	1	1	1	2	2	2	3	3	4	5	6	7
11.0	0	0	0	0	0	1	1	1	2	2	3	4	4	5	6	7
11.5	0	0	0	0	0	1	1	1	2	2	3	4	4	5	7	8
12.0	0	0	0	0	0	0	1	1	2	2	3	4	5	6	7	9
12.5	0	0	0	0	0	0	1	1	2	2	3	4	5	6	8	10
13.0	0	0	0	0	0	0	0	1	2	2	3	4	5	7	9	11
13.5	0	0	0	0	0	0	0	1	2	2	3	4	6	7	9	12
14.0	-1	0	0	0	0	0	0	1	2	2	4	5	6	8	10	13
14.5	-1	-1	-1	0	0	0	0	1	2	3	4	5	7	9	12	15
15.0	-1	-1	-1	0	0	0	0	1	2	3	4	6	8	10	13	16
15.5	-1	-1	-1	0	0	0	0	1	2	3	5	6	9	11	14	18
16.0	-1	-1	-1	-1	0	0	0	1	2	4	5	7	10	13	16	21
16.5	-1	-1	-1	-1	0	0	0	1	2	4	6	8	11	14	18	23
17.0	-1	-1	-1	-1	0	0	0	1	3	4	6	9	12	15	20	26
17.5	-1	-1	-1	-1	0	0	1	2	3	5	7	10	13	17	22	29
18.0	-2	-1	-1	-1	0	0	1	2	4	5	8	11	14	19	25	33
18.5	-2	-1	-1	-1	0	0	1	2	4	6	9	12	16	21	28	36
19.0	-2	-1	-1	-1	0	0	1	3	4	7	10	13	18	24	31	41
19.5	-2	-1	-1	0	0	0	1	3	5	7	11	15	20	26	35	46
20.0	-2	-1	-1	0	0	0	2	3	6	8	12	16	22	29	38	51
20.5	-1	-1	-1	0	0	1	2	4	6	9	13	18	24	32	43	57
21.0	-1	-1	-1	0	0	1	3	4	7	10	14	19	26	35	47	63
21.5	-1	-1	-1	0	0	1	3	5	8	11	15	21	29	39	52	70
22.0	-1	-1	0	0	0	2	3	6	8	12	17	23	32	43	58	78
22.5	-1	-1	0	0	1	2	4	6	9	13	18	25	34	47	64	87
23.0	-1	-1	0	0	1	2	4	7	10	14	20	28	38	51	70	96
23.5	-1	0	0	0	1	3	5	7	11	16	22	30	41	56	78	107
24.0	-1	0	0	0	1	3	5	8	12	17	24	33	45	62	85	119

TABLE II

```
      SUBROUTINE FAST16(WAVE,NX,NT,MODE,S,U,V,E,F)
      REAL WAVE(NX,NT),S(NX),U(NX),V(NX),E(NX),F(NX)
      REAL T(3,3),R(3),X(3),MU,NU
      LOGICAL QBC/T/
      DATA A2/2.0/,APB/1.148679/,ADIAG/2.70264/,AOFF/-.851321/,
     1 ALPHA/-.621556E-1/,BETA/.193140/,MU/.351735E-1/,
     2  NU/.838425E-1/,GAMMA/.598396E-2/,ZETA/.151982/
C
      NX1=NX-1
      NBASE=0
      IF (MODE.EQ.-1) NBASE=NT+1
C
C  ZERO TIME INTEGRATION BUFFERS.
C
      DO 10 I=1,NX
   10 U(I)=0.
      SUML =0.
      SUMR=0.
      TERML=0.
      TERMR=0.
C
C  LOOP OVER X-T FRAME, PROCESSING ONE X STRIP AT A TIME
C
      DO 50 JT=1,NT
      J= NBASE +MODE*JT
C
C  COMPUTE RHS OF INTERIOR WAVE EQUATION, AND RHS TERMS OF B.C.'S.
C
      DO 20 I=1,NX
      S(I)= WAVE(I,J)
   20 E(I)= APB*S(I) + A2*U(I)
      DO 30 I=2,NX1
   30 V(I)=S(I) +E(I-1) -2.*E(I) +E(I+1)
      PL=S(1)
      QL=S(2)
      PR=S(NX1)
      QR=S(NX)
      RHSL = NU*PL +MU*QL +ZETA*SUML+GAMMA*TERML
      RHSR = MU*PR +NU*QR -ZETA*SUMR+GAMMA*TERMR
C
C  SOLVE TRIDIAGONAL SYSTEM WITH OR WITHOUT ABSORBING SIDES.
C
C
      IF (QBC) CALL TRIBC(AOFF,ADIAG,AOFF,NX,WAVE(1,J),V,E,F,
     *          BETA,ALPHA,RHSL,ALPHA,BETA,RHSR)
C.....SOLVE TRIDIAGONAL SYSTEM WITH OLD B.C.'S.
      IF (.NOT.QBC) CALL TRIBC(AOFF,ADIAG,AOFF,NX,WAVE(1,J),
     * V,E,F,   1.0, -1.0, 0.0,  1.0,-1.0, 0.0)
C  UPDATE TIME INTEGRATIONS.
C
      DO 40 I=1, NX
   40 U(I) = U(I) +S(I) +WAVE(I,J)
      SUML = SUML +WAVE(2,J) -WAVE(1,J) -QL +PL
      SUMR = SUMR +WAVE(NX,J) -WAVE(NX1,J) -QR +PR
      TERML= WAVE(1,J) +WAVE(2,J) +QL +PL -TERML
      TERMR= WAVE(NX1,J) +WAVE(NX,J) +QR +PR -TERMR
C
```

```
C  END OF MAIN LOOP.
C
   50 CONTINUE
      RETURN
C
C  SPECIAL ENTRY TO SET THE CONSTANTS FOR THE MIGRATION
C
C  AND BOUNDARY CONDITIONS.
C  WMAX= MAXIMUM FREQUENCY OF DATA FOR FITTING B.C.'S.
C  X1,X2,X3= FITTING POINTS BETWEEN INTERIOR WAVE EQUATION
C             AND BOUNDARY CONDITION. (VALUES OF KX*W/VEL).
C  B= CONSTANT OF MODIFIED 15 DEGREE WAVE EQUATION.
C  IBC= 1=> USE ABSORBING SIDE B.C.'S.
C       0=> USE REFLECTING SIDES.
C
      ENTRY F16SET(DX,DT,DZ,VEL,WMAX,X1,X2,X3,B,IBC)
      QBC=.TRUE.
      IF (IBC.EQ.0) QBC=.FALSE.
      A= DZ*DT*VEL/(8.*DX**2)
      A2= 2.*A
      APB= A+B
      ADIAG= 1. +A2 -2.*B
      AOFF= B-A
      WRITE(6,100) DX,DT,DZ,VEL,A,B
  100 FORMAT(1HO,'FAST16 PARAMETERS:  DX=',F8.4,'  DT=',F8.4,
     1 '  DZ=',F8.4,'  VEL=',F9.4,'  A=',F9.5,'  B=',F9.6)
      IF (.NOT.QBC) RETURN
C
C  CALCULATE B.C. COEFFICIENTS.
C
C  SET UP SYSTEM T*C=R TO SOLVE FOR (C1,C2,C3), THE CONSTANTS
C  OF THE TRANSFORMED B.C. DISPERSION RELATION.  THE FITTING
C  POINTS ARE (X1,X2,X3)
C
      PER=0.5*WMAX*DX/VEL
      X(1)= X1
      X(2)= X2
      X(3)= X3
      DO 60 I=1,3
      TANG= TAN( X(I)*PER )
      SIN2= (SIN(X(I)*PER))**2
      T(1,I)=  1.0 -4.*B*SIN2
      T(2,I)= -TANG/PER*T(1,I)
      T(3,I)= -0.5*TANG*SIN2/PER**3
   60 R(I)  = -0.5*SIN2/PER**2
C
C  SOLVE SYSTEM.
C
      DET=T(1,1)*(T(2,2)*T(3,3)-T(2,3)*T(3,2))-T(2,1)*(T(1,2)*T(3,3)
     * -T(1,3)*T(3,2))+T(3,1)*(T(1,2)*T(2,3)-T(1,3)*T(2,2))
      C1=(R(1)*(T(2,2)*T(3,3)-T(2,3)*T(3,2))-T(2,1)*(R(2)*T(3,3)-R(3)
     * *T(3,2))+T(3,1)*(R(2)*T(2,3)-R(3)*T(2,2)))/DET
      C2= (T(1,1)*(R(2)*T(3,3)-R(3)*T(3,2))-R(1)*(T(1,2)*T(3,3)-T(1,3)
     * *T(3,2))+T(3,1)*(T(1,2)*R(3)-T(1,3)*R(2)))/DET
      C3=(T(1,1)*(T(2,2)*R(3)-T(2,3)*R(2))-T(2,1)*(T(1,2)*R(3)-T(1,3)
     * *R(2))+R(1)*(T(1,2)*T(2,3)-T(1,3)*T(2,2)))/DET
C
```

```
C  COMPUTE CONSTANTS OF B.C.
C
      ETA= C2/DX
      ZETA= (C3*VEL*DT)/(DZ*DX)
      GAMMA= C1/(VEL*DT)
      ALPHA= (1./DZ -ZETA -ETA +GAMMA)/2.
      BETA = (1./DZ +ZETA +ETA +GAMMA)/2.
      MU   = (1./DZ -ZETA +ETA -GAMMA)/2.
      NU   = (1./DZ +ZETA -ETA -GAMMA)/2.
C
      WRITE(6,101) X
  101 FORMAT(2X,'BOUNDARY CONDITIONS FITTED AT KX*VEL/W=',3G16.7)
      WRITE(6,102) C1,C2,C3
  102 FORMAT(2X,'C1,C2,C3=',3G16.7)
      WRITE(6,103) ALPHA,BETA,MU,NU,ETA,ZETA,GAMMA,DET
  103 FORMAT(2X,'ALPHA=',G16.7,'  BETA=',G16.7,'  MU=',G16.7,
     1 '  NU=',G16.7,/,2X,'  ETA=',G16.7,'  ZETA=',G16.7,
     2 '  GAMMA=',G16.7,'  DET=',G16.7)
      RETURN
      END
      SUBROUTINE TRIBC(AL,AD,AU,N,T,D,E,F,A1,A2,A3,B1,B2,B3)
      REAL T(N),D(N),E(N),F(N)
C
C  THE SAME AS SUB/R TRI EXCEPT BOUNDARY CONDITIONS OF THE
C  FOLLOWING FORM ARE USED.
C
C     A1*T(1) + A2*T(2) = A3
C     B1*T(NX-1) + B2*T(NX) = B3
C
      N1 = N-1
      E(1) = -A2/A1
      F(1) = A3/A1
      DO 10 I=2,N1
      DEN = AD + AL*E(I-1)
      E(I) = -AU/DEN
   10 F(I) = (D(I) - AL*F(I-1)) / DEN
      T(N) = ( B3 - B1*F(N1) )/( B1*E(N1) + B2 )
      DO 20 J=1,N1
      I=N-J
   20 T(I) = E(I)*T(I+1) + F(I)
      RETURN
      END
```